

### Math 2280 Maple Project 3: Laplace Applications S2015

Maple lab 3 has five problems L3-1, L3-2, L3-3, L3-4, L3-5.

**References:** Code in maple appears in 2280mapleL3-laplace-S2015.txt at URL <http://www.math.utah.edu/~gustafso/>. This document: 2280mapleL3-laplace-S2015.pdf. Other related and required documents are available at the web site.

#### Problem L3-1. (Periodic Wave Plots)

In the table are examples of standard periodic waves used in engineering applications. The last table column has piecewise expressions  $h$  defined on the base interval  $[0, T]$ .

Given an expression  $h$  on the base  $[0, T]$ , then its  $T$ -periodic extension  $H$  to  $(-\infty, \infty)$  is always  $H(x) = h(g(x))$  where  $g(x) = x - T \text{ floor}(x/T)$ .

In terms of the *triangular wave*  $\text{twave}(x) = x - \text{floor}(x)$ , we may write  $g(x) = T \text{ twave}(x/T)$ . The triangular wave is remembered as a  **$T$ -periodic train of ramps**. The *staircase function*  $\text{floor}(x)$  used in the construction of the ramp train is **not periodic**; it is a standard math library function in major programming languages.

(a) Plot all the periodic examples **f1** to **f7**. Please choose an appropriate graph window for each.

(b) Justify every maple expression in the table. Use the example below as a guide.

**Example.** To justify the first table entry, which contains math definition **f1(x)** and piecewise definition **h1(x)** for the square wave, define **H1(x)** to be the  $T$ -periodic extension of **h1(x)** to the whole real line, then plot **H1(x)-f1(x)** over three periods. The plot should be the zero function, i.e., the  $x$ -axis.

```
# Maple details for the example
opt1:=ytickmarks=3,color=red,labels=[x,'f(x)'],title="square wave";
opt2:=numpoints=100,thickness=2,discont=true;
opts:=opt1,opt2;
f1:=x -> (-1)^(floor(x));
T:=2; # T = period = 2
g:=x -> x-T*floor(x/T);
h1:=x -> piecewise(x<1,1,x<2,-1,0);
H1:=x -> h1(g(x)); # T-periodic extension
plot(H1(x)-f1(x),x=0..3*T,opts); # Should plot as y=0 (the x-axis)
```

Maple Expression	Name	T	Piecewise Definition on $[0, T]$
<b>f1:=x -&gt; (-1)^(floor(x));</b>	square wave	2	$h_1(x) = \begin{cases} 1 & 0 \leq x < 1, \\ -1 & 1 \leq x < 2 \end{cases}$
<b>f2:=x -&gt; x-floor(x);</b>	triangular wave	1	$h_2(x) = \begin{cases} x & 0 \leq x < 1, \\ 0 & x = 1 \end{cases}$
<b>f3:=x -&gt; 1/2+(f2(x)-1/2)*f1(x);</b>	sawtooth wave	2	$h_3(x) = \begin{cases} x & 0 \leq x < 1, \\ 2-x & 1 \leq x < 2 \end{cases}$
<b>f4:=x-&gt;abs(sin(x));</b>	rectified sine	$2\pi$	$h_4(x) = \begin{cases} \sin(x) & 0 \leq x < \pi, \\ -\sin(x) & \pi \leq x < 2\pi \end{cases}$
<b>f5:=x-&gt;(sin(x)+abs(sin(x)))/2;</b>	half-wave rectified sine	$2\pi$	$h_5(x) = \begin{cases} \sin(x) & 0 \leq x < \pi, \\ 0 & \pi \leq x < 2\pi \end{cases}$
<b>p:= x -&gt; (2-x)*x;</b> <b>f6:=t-&gt;p(2*f2(x/2))*f1(x/2);</b>	parabolic wave	4	$h_6(x) = \begin{cases} p(x) & 0 \leq x < 2, \\ -p(x-2) & 2 \leq x < 4 \end{cases}$
<b>q:=x-&gt;</b> <b>piecewise(x&lt;Pi,sin(x),x&lt;2*Pi,-1):</b> <b>f7:=x-&gt;q(2*Pi*f2(x/2/Pi));</b>	piecewise sine pulse	$2\pi$	$h_7(x) = \begin{cases} \sin(x) & 0 \leq x < \pi, \\ -1 & \pi \leq x < 2\pi \end{cases}$

### Problem L3-2. (Hammer Hit Oscillation)

An attached mass in an undamped spring-mass system is released from rest 1 meter below the equilibrium position. After 3 seconds of oscillation, the mass is struck by a hammer with force of 5 Newtons in a downward direction.

(a) Assume the model

$$\frac{d^2x}{dt^2} + 9x = 5\delta(t-3); x(0) = 1, \frac{dx}{dt}(0) = 0,$$

where  $x(t)$  denotes the displacement from equilibrium at time  $t$  and  $\delta(t-3)$  denotes the Dirac delta function. Determine, using the `dsolve` example below, a piecewise-defined formula for  $x(t)$ . Plot  $x(t)$  for  $0 \leq t \leq 7$ .

(b) Solve the following hammer-hit models DE1 to DE4, given as maple expressions, using the `dsolve` example for DE, IC as a template for the solution.

(c) Express the symbolic answer for each of DE1 to DE4 as a piecewise-defined function. Interpret each answer physically from the initial conditions and the applied impulse.

```
DE:=diff(x(t),t,t)+9*x(t)=3*Dirac(t-3); IC:=x(0)=1,D(x)(0)=0;
dsolve({DE,IC},x(t),method=laplace);
# x(t) = cos(3*t)+Heaviside(t-3)*sin(-9+3*t)
convert(%,piecewise);combine(%,trig);
# x(t) = cos(3*t) for t < 3,cos(3*t)+sin(-9+3*t) for t>3, undef t=3.
```

```
DE1:=diff(x(t),t,t)+9*x(t)=5*Dirac(t-3); IC1:=x(0)=-1,D(x)(0)=1;
DE2:=diff(x(t),t,t)+9*x(t)=6*Dirac(t-3); IC2:=x(0)=1,D(x)(0)=-1;
DE3:=diff(x(t),t,t)+9*x(t)=8*Dirac(t-3); IC3:=x(0)=0,D(x)(0)=-1;
DE4:=diff(x(t),t,t)+9*x(t)=9*Dirac(t-3); IC4:=x(0)=1,D(x)(0)=0;
```

### Problem L3-3. (Maple Solution of Initial Value Problems)

(a) Solve the IVP  $y'' - y' - 2y = 5 \sin x$ ,  $y(0) = 1$ ,  $y'(0) = -1$ . Please use the `inttrans` package. Show the steps in Laplace's method, entirely in maple, with explicit use of maple functions `laplace(f,t,s)` and `invlaplace(F,s,t)`. The solution should duplicate the major steps that would be done by hand, table details omitted.

(b) Solve the pulse-input IVP

$$3y'' + 3y' + 2y = \begin{cases} 0 & \text{for } t < 0, \\ 3 & \text{for } 0 \leq t < 4, \\ 0 & \text{for } t \geq 4, \end{cases}$$

with initial data  $y(0) = 0$ ,  $y'(0) = 0$ . Use any maple method. Express your answer as a piecewise-defined function.

(c) Solve the IVP  $y'' + y = 1 + \delta(t-2\pi)$ ,  $y(0) = 1$ ,  $y'(0) = 0$ . Use maple `dsolve`. Express the answer as a piecewise-defined function.

### Problem L3-4. (Expressions for Periodic Waves)

Let  $h$  be the  $T$ -periodic extension to  $-\infty < x < \infty$  of  $f(x)$ , which is only defined on  $0 \leq x \leq T$ . Define  $T = 2$  and  $f(x) = 2/10 + (7/10) \sin x + (1/10) \cos 5x$  on  $[0, T]$ .

(a) Plot  $h(t)$  on the interval  $[-10, 10]$ . Use the composition formula  $h(t) = f(g(t))$ , where  $g(t) = t - T \text{ floor}(t/T)$ .

(b) Compute the Laplace of  $h(t)$  directly from the periodic function theorem, using the sample maple code

```
int(f(g(t))*exp(-s*t),t=0..T)/(1-exp(-s*T));
```

Replacing  $f(x)$  by  $(1/10) \cos(5x)$  should give the answer below. The answer for  $2/10 + (7/10) \sin x + (1/10) \cos 5x$  has many more terms.

$$\frac{1}{10} \frac{se^{2s} - s \cos(10) + 5 \sin(10)}{(s^2 + 25)(-1 + e^{2s})}$$

(c) Maple directly finds the laplace of  $g(t) = t - T \text{ floor}(t/T)$ , but not the laplace of  $h(t) = f(g(t))$ . Truncating  $f(x) = \frac{2}{10} + \frac{7}{10} \sin(x) + \frac{1}{10} \cos(5x)$  to the constant term  $2/10$  allows maple to compute the Laplace of  $f(g(t))$ . But the sine and cosine terms do not evaluate.

To get help from maple, the function  $h(t)$  is expressed as a series of pulses. The laplace of the series  $h(t)$  can be computed, provided  $\frac{1}{10} \cos(5x)$  is removed from  $f(x)$ . This example shows that the periodic function theorem is a basic tool in Laplace theory. Here's the success story for this example:

```

pulse:=(t,a,b)->Heaviside(t-a)-Heaviside(t-b);
f := x -> 2/10+7/10*sin(x): h:= t->sum(f(t-n*T)*pulse(t,n*T,n*T+T),n=0..infinity);
intttrans[laplace](h(t),t,s);
eval(%) assuming n::positive;

```

Type this code into maple and report the answer. Check the answer by comparing terms in the solution to part (b) above.

**REMARK.** Here's what does not work. Beware of testing the code below: it uses about 800mb memory and may finish with no answer. If you find a way to resolve the difficulty for all versions of maple, then please send email, detailing how to do it.

```

pulse:=(t,a,b)->Heaviside(t-a)-Heaviside(t-b);
f := x -> (1/10)*cos(5*x):
h:= t->sum(f(t-n*T)*pulse(t,n*T,n*T+T),n=0..infinity);
intttrans[laplace](h(t),t,s);
eval(%) assuming n::positive;

```

### Problem L3-5. (Resolvent Method)

The Laplace resolvent formula for the problem  $\mathbf{u}' = A\mathbf{u}$ ,  $\mathbf{u}(0) = \mathbf{u}_0$  is

$$\mathcal{L}(\mathbf{u}(t)) = (sI - A)^{-1}\mathbf{u}_0.$$

For example,  $A = \begin{pmatrix} 1 & 0 \\ 0 & 2 \end{pmatrix}$  gives

$$\mathcal{L}(\mathbf{u}(t)) = \begin{pmatrix} s-1 & 0 \\ 0 & s-2 \end{pmatrix}^{-1} \mathbf{u}_0 = \begin{pmatrix} \frac{1}{s-1} & 0 \\ 0 & \frac{1}{s-2} \end{pmatrix} \mathbf{u}_0 = \begin{pmatrix} \mathcal{L}(e^t) & 0 \\ 0 & \mathcal{L}(e^{2t}) \end{pmatrix} \mathbf{u}_0,$$

which implies  $\mathbf{u}(t) = \begin{pmatrix} e^t & 0 \\ 0 & e^{2t} \end{pmatrix} \mathbf{u}_0$ .

The answers for the components of  $\mathbf{u}$  are  $\alpha e^t$ ,  $\beta e^{2t}$ , according to the following maple code:

```

with(LinearAlgebra):with(intttrans):
A:=Matrix([[1,0],[0,2]]);
u0:=Vector([alpha,beta]);
B:=(s*IdentityMatrix(2)-A)^(-1).u0;
u:=Map(invlaplace,B,s,t);

```

Compute the solution  $\mathbf{u}(t)$  using the resolvent formula for the following cases.

(a)  $A = \begin{pmatrix} 2 & 1 \\ 1 & 2 \end{pmatrix}$ ,  $\mathbf{u}(0) = \begin{pmatrix} 0 \\ -1 \end{pmatrix}$

(b)  $A = \begin{pmatrix} 0 & 1 \\ -1 & -2 \end{pmatrix}$ ,  $\mathbf{u}(0) = \begin{pmatrix} 1 \\ -1 \end{pmatrix}$

(c)  $A = \begin{pmatrix} 2 & -5 \\ 4 & -2 \end{pmatrix}$ ,  $\mathbf{u}(0) = \begin{pmatrix} -1 \\ 2 \end{pmatrix}$